


Проект
(Project1.vbp)


Библиотека (модул) с
инструменти за ЕГН
(modEGN.bas)

```
Function BDate(ByVal s As String) As Date
As Date
Function TrueEgn(ByVal s As String) As Boolean
Function Gender(ByVal s As String) As Byte
As Byte
```

Форма за тестване
(Form1.frm)



Потребителски контрол
(UCEGN.ctl)



```
Event Change()
Event Correct()
Property Text (published)
Property Gender (Read-only)
Property BirthDate (Read-only)
Method: Function TrueEgn() As Boolean
```

(Form1)

```
Option Explicit
Option Explicit
Private Declare Function Beep _
    Lib "kernel32" (ByVal dwFreq As Long, _
    ByVal dwDuration As Long) As Long
Private Sub UCEgn1_Change()
    Select Case UCEgn1.Gender
        Case 0: LGenger.Caption = ""
        Case 1: LGenger.Caption = "M"
        Case 2: LGenger.Caption = "F"
    End Select
    If UCEgn1.BirthDate <> "" Then
        LBDate.Caption = UCEgn1.BirthDate
    Else
        LBDate.Caption = ""
    End If
End Sub
Private Sub UCEgn1_Correct()
    Dim i As Integer
    For i = 500 To 3000 Step 200
        Beep i, 10
    Next
End Sub
(modEGN)
Option Explicit
Function BDate(ByVal s As String) As Date
Dim d As Integer, m As Integer, g As Integer
Dim dt As Date
BDate = 0
If Len(s) < 6 Then Exit Function
d = Mid(s, 5, 2)
m = Mid(s, 3, 2)
g = Mid(s, 1, 2)
Select Case m
    Case 1 To 12
        g = 1900 + g
    Case 21 To 32
        g = g + 1800
        m = m - 20
    Case 41 To 52
        g = g + 2000
        m = m - 40
    Case Else
        Exit Function
    End Select
dt = DateSerial(g, m, d)
If d <> Day(dt) Or m <> Month(dt) Or g <> Year(dt)
Then Exit Function
BDate = dt
End Function
```

```
Function TrueEgn(ByVal s As String) As Boolean
Dim i As Integer, p As Integer, d As Integer
TrueEgn = False
If Len(s) <> 10 Then Exit Function
p = 0
On Error GoTo err
For i = 9 To 1 Step -1
    d = Mid(s, i, 1)
    p = (2 * p + d) Mod 11
Next
If BDate(s) = 0 Then Exit Function
d = Mid(s, 10, 1)
TrueEgn = ((2 * p) Mod 11) Mod 10 = d
err:
End Function
Function Gender(ByVal s As String) As Byte
If TrueEgn(s) Then
    Gender = 1 + Mid(s, 9, 1) Mod 2
Else
    Gender = 0
End If
End Function
(UCEGN)
Option Explicit
Event Change()
Event Correct()
Private Sub TEgn_Change()
    If TrueEgn Then
        ShEgn.FillColor = vbGreen
        ShEgn.BorderColor = vbGreen
        RaiseEvent Correct
    Else
        ShEgn.FillColor = vbRed
        ShEgn.BorderColor = vbRed
    End If
    RaiseEvent Change
End Sub
Private Sub TEgn_KeyPress(KeyAscii As Integer)
    If (KeyAscii < vbKey0 Or KeyAscii > vbKey9) And
    KeyAscii <> vbKeyBack Then
        KeyAscii = 0
    End If
End Sub
Public Property Get Text() As String
    Text = TEgn.Text
End Property
Public Property Let Text(ByVal vNewValue As String)
    If modEGN.TrueEgn(vNewValue) Then
        TEgn.Text = vNewValue
        PropertyChanged "Text"
    Else
        err.Raise vbObjectError + 1, , "Wrong EGN"
    End If
End Property
Public Function TrueEgn() As Boolean
    TrueEgn = modEGN.TrueEgn(TEgn.Text)
End Function
Property Get Gender() As Byte
    Gender = modEGN.Gender(TEgn.Text)
End Property
Property Get BirthDate() As String
    Dim dt As Date
    dt = BDate(TEgn.Text)
    If dt = 0 Then
        BirthDate = ""
    Else
        BirthDate = Format(dt, "dd.mm.yyyy")
    End If
End Property
Private Sub UserControl_ReadProperties _
    (PropBag As PropertyBag)
    TEgn.Text = PropBag.ReadProperty("Text", "")
End Sub
Private Sub UserControl_Resize()
    TEgn.Width = ScaleWidth - ScaleHeight
    TEgn.Height = ScaleHeight
    TEgn.Left = ScaleHeight
    Shape2.Height = TEgn.Height
    Shape2.Width = Shape2.Height
    ShEgn.Height = TEgn.Height
    ShEgn.Width = ShEgn.Height
End Sub
Private Sub UserControl_WriteProperties _
    (PropBag As PropertyBag)
    PropBag.WriteProperty "Text", TEgn.Text, ""
End Sub
```